Evaluation of Pr $\{x \geq y\}$ When Both X and Y are from Three-Parameter Weibull Distributions

GERALD G. BROWN AND HERBERT C. RUTEMILLER

Abstract–It is important in many reliability applications to determine the probability that the failure time of an element from one population will exceed that of an element from a second population. In this paper, we present a method for computer calculations of Pr $\{x \geqslant y\}$ where X and Y are each from a three-parameter Weibull distribution. In addition, we provide the moments and the probability density function of the difference. Numerical examples are included.

Reader Aids:
Purpose: Report of software development
Special math needed for explanations: Statistics, numerical analysis
Special math needed for results: same
Results useful to: Statisticians, theoretically inclined reliability engineers

1. INTRODUCTION

The Weibull cumulative distribution function [8],

$$F(x) = 1 - \exp\{-(x-\gamma)^{\beta}/\alpha\}, \quad \gamma \leqslant x \leqslant \infty, \alpha, \beta > 0,$$

has been widely used as a model for the failure distribution of components under mechanical or electrical loading. A variety of techniques is available in the literature for the estimation of the three Weibull parameters, α, β, γ, from ordered or unordered sample observations. Ravenis [7] presents an excellent summary of estimation methods based upon completely random samples, while Mann [5] has discussed estimation from censored samples. The methods suggested by these authors for finding maximum likelihood or moment estimators require iterative, computer solutions. Berrettoni [2] and Nelson and Thompson [6] suggest a graphical solution through the use of special Weibull probability paper.

Suppose, now, that we have life test data from two distinct populations, X and Y, and have estimated Weibull parameters (α, β, γ) and $(\alpha^*, \beta^*, \gamma^*)$, respectively. We consider in this paper the problem of estimating Pr $\{x \geqslant y\}$. This estimator may be required in the following circumstances:

(a) The random variables, X and Y, represent the fatigue life, in cycles, for a cyclically stressed component made from two alternate materials of construction, say an aluminum alloy and a steel alloy. If we simply wish to maximize the reliability of the component at n_0 cycles, then a choice can be made by calculating

$$R_x = \exp\{-(n_0-\gamma)^{\beta}/\alpha\},$$
$$R_y = \exp\{-(n_0-\gamma^*)^{\beta^*}/\alpha^*\},$$

and choosing the higher reliability. However, if we want to design as long-lived a product as possible, we should calculate, instead, Pr $\{x \geqslant y\}$ and choose X or Y when this probability is above or below 0.5.

(b) A second application involves "interference" problems where the random variable, X, represents a strength distribution and the random variable, Y, the distribution of service stress. A failure occurs if Y exceeds X, so, in this instance, the reliability of the system is precisely Pr $\{x \geqslant y\}$.

(c) A complex system with two components in series possessing independent Weibull failure distributions can fail as a consequence of failure of component X or component Y. Suppose that one is a "safe" mode of failure, and the other a "dangerous" mode. For example, in an exhaust fan installation, Y might represent failure of the bearing while X might represent fatigue failure of a blade. Then Pr $\{x \geqslant y\}$ is the probability of a safe mode of failure.

2. METHODS OF SOLUTION

If X and Y are statistically independent Weibull variables with known, or estimated parameters, then

$$\Pr\{x \geqslant y\} = \int_0^{\infty} f_y(t; \alpha^*, \beta^*, \gamma^*) R_x(t; \alpha, \beta, \gamma)\, dt, \quad (1)$$

where f_y = pdf $\{y\}$ and $R_x = 1 -$ cdf $\{x\}$.

This expression cannot be integrated analytically. Possible methods for evaluating (1) include the following:

1. Numerically integrate (1) directly.
2. Obtain the pdf for the random variable $W \equiv X - Y$ by first performing a transformation from the joint pdf, $f(x, y)$, to the joint pdf, $h(x, w)$. Find the marginal pdf for W by integration over X, and integrate the marginal pdf of W over $(0, \infty)$.
3. Find the Laplace Transform,

$$l\{w\} = l\{x\}\, l\{-y\}.$$

Invert $l\{w\}$ to obtain pdf $\{w\}$, and integrate W over $(0, \infty)$.

4. Find the moments of W in terms of the moments of X and Y. Then fit an appropriate pdf for W by the method of

Manuscript received January 17, 1972; revised August 30, 1972.
The authors are with the Department of Quantitative Methods, California State University, Fullerton, Calif. 92631.

moments, and integrate over $(0, \infty)$. The k^{th} moment of the Weibull distribution [4] is:

$$E\{x - \gamma\}^k = \alpha^{k/\beta}\, \Gamma\left(\frac{k}{\beta} + 1\right).$$

For example, the mean and variance of the difference of two independent Weibulls are–

$$E\{w\} = \alpha^{1/\beta}\, \Gamma\left(\frac{1}{\beta} + 1\right) - \alpha^{*1/\beta}\, \Gamma\left(\frac{1}{\beta^*} + 1\right) + \gamma - \gamma^*; \quad (2)$$

$$\operatorname{var}\{w\} = \alpha^{2/\beta}\left\{\Gamma\left(\frac{2}{\beta} + 1\right) - \Gamma^2\left(\frac{1}{\beta} + 1\right)\right\} + \alpha^{*2/\beta^*}\left\{\Gamma\left(\frac{2}{\beta^*} + 1\right) - \Gamma^2\left(\frac{1}{\beta^*} + 1\right)\right\}. \quad (3)$$

We have chosen the first of these four methods. Method 2 leads to a numerical integration problem comparable to that for direct integration. Method 3 is difficult to apply because the Laplace Transform of the Weibull distribution is not available in closed form. Numerical inversion, or algebraic inversion with continued fractions, is necessary, and the method leads to more complicated numerical analysis difficulties than direct integration. Method 4 provides, at best, a rough approximation to the pdf of the difference. The problem is particularly difficult because the domain for W is $(-\infty, \infty)$, and the distribution will certainly be asymmetric for most parameter combinations. A detailed examination of the shape of this family of distributions would be necessary before selecting a suitable function to fit them, as required by Method 4.

3. SOLUTION

Selection of a method for direct numerical integration of (1) should take advantage of the analytically known integrand but recognize the cost of each evaluation of this complicated function; the Gauss-Legendre quadrature method [3] was investigated because of its accuracy and low computation cost. This method [1, Sec. 25] approximates an integration by

$$\int_{-1}^{1} g(s)\, ds \simeq \sum_{i=0}^{n} w_i\, g(s_i) \quad (4)$$

where w and s are, respectively, weights and evaluation base points tabulated for interpolation order $n = 2, 3, \cdots$. This is very easy to program for machine use and has excellent theoretical error terms for exponential forms such as (1).

Restating (1) in the form of (4) requires either a Jacobian transformation, such as

$$s = 1 - 2\exp(-t),$$

or selection of a heuristic finite upper integration bound. The latter method was found to be superior numerically to the former, possibly due to the added demands of the transformed problem on computational precision.

The heuristic method was used to restate (1) as

$$\Pr\{x \geqslant y\} \simeq \int_{\gamma^*}^{b} f_y(t; \alpha^*, \beta^*, \gamma^*)\, R_x(t; \alpha, \beta, \gamma)\, dt, \quad (5)$$

with

$$b \equiv R_y^{-1}(\xi) = (-\alpha^* \log \xi)^{1/\beta^*} + \gamma^*,$$

$$\xi = \text{reliability arbitrarily close to zero.}$$

This amounts to directing the attention of the numerical integration to an interesting domain where $f_y \neq 0$. Further, the error introduced by such truncation is

$$\int_b^\infty f_y(t) R_x(t)\, dt \leqslant \int_b^\infty f_y(t)\, dt = R_y(b) = \xi.$$

The value of ξ was taken to be 10^{-6}. Because of the sensitivity of the Weibull pdf to changes in its parameters, especially the shape parameter β, this method works very well in practice.

It is also possible to generate numerically the difference of two Weibulls by method 2, where

$$h(x, w) \equiv f_y(x - w; \alpha^*, \beta^*, \gamma^*) f_x(x; \alpha, \beta, \gamma), \quad \gamma \leqslant x \leqslant \infty,\ -\infty \leqslant w \leqslant \infty.$$

Then the pdf for W is

$$g(w) \equiv \int_a^\infty h(x, w)\, dx \simeq \int_a^b h(x, w)\, dx, \quad -\infty \leqslant w \leqslant \infty,$$

where

$$a \equiv \begin{cases} w + \gamma^*, & \text{for } w > \gamma - \gamma^* \\ \gamma, & \text{otherwise.} \end{cases}$$

As before, a heuristic selection of b was used.

$$b \equiv \max\{R_y^{-1}(\xi), R_x^{-1}(\xi)\}.$$

4. NUMERICAL EXAMPLES

(a) A special case of the Weibull distribution is the two-parameter exponential distribution, obtained by setting the parameter, $\beta = 1$. When $\beta = \beta^* = 1$, it is possible to find $\Pr\{x \geqslant y\}$ analytically by method 2. The solution is

$$\Pr\{x \geqslant y\} = \frac{\alpha}{\alpha + \alpha^*} \exp -\{(\gamma^* - \gamma)/\alpha\}, \quad \text{if } \gamma^* \geqslant \gamma. \quad (6)$$

If $\gamma^* < \gamma$, interchange the definition of X and Y. For example, life tests were performed on two alternate designs of an electronic circuit, where the distribution of failure times was assumed to be 2 - parameter exponential. The MTBF and first failure times for designs X and Y were estimated as (1800, 0) and (1600, 300) hours respectively. A numerical integration using the 20 point Gauss-Legrende Formula indicated $\Pr\{x \geqslant y\} = 0.44813$. Analytically (6) gives as the exact solution $\Pr\{x \geqslant y\} = 0.44814$. The pdf of W obtained by similar numerical integration agreed to 5 significant digits over the domain $(-20, 20)$ with the exact solution:

$$h(w) = \begin{cases} \dfrac{1}{\alpha + \alpha^*} \exp\{-(w - \gamma + \gamma^*)/\alpha\}, & \text{for } w > \gamma - \gamma^*; \\[2ex] \dfrac{1}{\alpha + \alpha^*} \exp\{(w - \gamma + \gamma^*)/\alpha^*\}, & \text{otherwise.} \end{cases}$$



Fig. 1. Probability density function of X-Y for two-parameter exponential variates X and Y.

(b) Weibull distributions were fitted to life test data on two competing designs for bearings subjected to a constant radial load. With life expressed in millions of cycles, the estimated Weibull parameters were—

Design	α	β	γ
X	210.4	2.85	40.1
Y	309.4	1.15	0.0

The integration indicated $\Pr\{x \geqslant y\} = 0.232$, so that design Y should be preferred. We cannot, of course, verify this probability analytically. However, one check on the numerical accuracy of the method is available by reversing the definition of X and Y and performing the integration again on the asymmetric complementary argument. Exact results were:

$$\Pr\{x \geqslant y\} = 0.2316150,$$

$$\Pr\{y \geqslant x\} = 0.7683854,$$

$$\text{Sum} = 1.0000004.$$

Using method 2, the pdf of W was numerically calculated for the exponential and Weibull examples. Results are exhibited graphically in Figures 1 and 2.

5. ESTIMATION OF WEIBULL PARAMETERS

The methods described above require that numerical values of all Weibull parameters be available. A convenient method for estimating (α, β, γ) from sample observations is the "pseudo-least-squares" procedure. Define

$$p \equiv \ln \ln \{1/R_x(x)\},$$

$$q \equiv \ln (x - \gamma),$$

so that

$$p = -\ln \alpha + \beta q. \tag{7}$$

If the ordered observations in a sample of size n are $x_{(1)}, \cdots x_{(n)}$, then an unbiased estimate of the reliability at time



Fig. 2. Probability density function of X-Y for three-parameter Weibull variates X and Y.

$x_{(i)}$ is $i/(n+1)$. Estimate $\hat{\gamma} = x_{(1)}$, and find least squares estimates, $\hat{\alpha}$ and $\hat{\beta}$, from (7), using points $x_{(2)}, \cdots x_{(n)}$. Then search the interval, $(0, x_{(1)})$, for alternate choices of $\hat{\gamma}$, solving (7) each time. Select the set $(\hat{\gamma}, \hat{\alpha}, \hat{\beta})$ which minimizes the squared deviations of p about the regression line.

A nonparametric method for the estimation of $\Pr\{x \geqslant y\}$ from sample observations is available through the use of the Mann-Whitney statistic. Count the number of pairs of values in two samples of size n_x and n_y for which $x_i > y_j, i = 1, \cdots, n_x; j = 1, \cdots, n_y$. Divide this number by $n_x n_y$ to obtain an unbiased estimate of $\Pr\{x \geqslant y\}$.

The nonparametric method avoids the necessity of Weibull parameter estimation and numerical integration. It is of interest, therefore, to compare the efficiency of the Mann-Whitney estimator with a parametric type.

No theoretical information is available on the mathematical properties of either competing estimator for the Weibull distribution. Therefore, several Monte Carlo simulations were performed. In each simulation, one hundred distinct random samples, each consisting of ten (x, y) pairs were drawn from two Weibull distributions. The ten pairs of observations were utilized to estimate (α, β, γ) and $(\alpha^*, \beta^*, \gamma^*)$ by the pseudo-least-squares procedure, and then to estimate $\Pr\{x > y\}$. The Mann-Whitney estimate of $\Pr\{x > y\}$ was calculated from the same ten pairs. Finally, the mean squared error of the one hundred parametric and nonparametric estimates of reliability was calculated. The simulation was performed for the two examples discussed earlier in the paper, plus five additional Weibull distributions. Then the entire experiment was repeated, with individual samples of twenty, rather than ten, (x, y) pairs. Results are shown in Table 1.

The mean squared error of the parametric estimator was about 20 percent smaller for the two examples discussed previously. In every one of the fourteen simulations attempted, the parametric estimator was superior.

TABLE 1

Relative Mean Squared Error of Parametric and Nonparametric Estimators of Pr $\{x > y\}$

α	β	γ	α^*	β^*	γ^*	Exact Pr $\{x > y\}$	One Hundred Samples M.S.E. (Mann-Whitney)/M.S.E. (Parametric) $n = 10$	$n = 20$
1800.0	1.0	0.0	1600.0	1.0	300.0	0.448	1.18	1.21
210.4	2.85	40.1	309.4	1.15	0.0	0.232	1.34	1.19
250.0	1.75	20.0	250.0	1.0	20.0	0.079	1.36	1.21
250.0	2.00	50.0	300.0	1.0	20.0	0.136	1.63	1.31
250.0	1.50	0.0	300.0	2.25	50.0	0.151	1.03	1.12
200.0	1.50	50.0	250.0	1.25	20.0	0.479	1.12	1.27
250.0	1.50	50.0	250.0	1.25	20.0	0.508	1.11	1.09

6. DISCUSSION

A copy of a FORTRAN program which calculates Pr $\{x \geqslant y\}$, given (α, β, γ) and $(\alpha^*, \beta^*, \gamma^*)$, using method 1, and calculates numerical points on the pdf of $X - Y$, using method 2, is given in the Appendix. The program also computes the mean and variance of the difference, using method 4.

The procedures described in this paper can be easily adapted to problems where any of the pdf's are other than Weibull.

APPENDIX

A FORTRAN program for evaluation of Pr $\{x \geqslant y\}$ and ordinates of the pdf of $W = X - Y$ when X and Y are 3-parameter Weibull variates.

The program is written in Fortran IV and should be acceptable for IBM, CDC, Burroughs, etc. machines. The integration subroutine, XINT, may reference any acceptable library quadrature method; a FORTRAN IV function for Legendre quadrature is given in [3, p109]. The interval-halving [3, p90] should be used for the more unruly Weibull distributions, e.g., large β.

```
      COMMON AX,BX,CX, AY,BY,CY, W
      INTEGER OUTPUT
      EXTERNAL P, H
      INPUT = 5
      OUTPUT = 6
C
C   XINT IS LIBRARY INTEGRATION FUNCTION
C   GAMMA IS LIBRARY GAMMA FUNCTION
C   TOL IS HEURISTIC ZERO LEVEL
C
      TOL = 0.000001
   10 READ(INPUT,20) AX,BX,CX, AY,BY,CY, NW, Z
   20 FORMAT( 6F5.0, I5, F5.0 )
      IF( AX .LE. 0.0 .OR. AY .LE. 0.0 ) STOP
      IF( BX .GT. 10.0 .OR. BY .GT. 10.0 ) STOP
C
C   MOMENTS OF DENSITIES, DENSITY OF DIFFERENCE
C
      EX = AX**(1.0/BX) * GAMMA(1.0/BX+1.0) + CX
      VX = AX**(2.0/BX) * (GAMMA(2.0/BX+1.0) - GAMMA(1.0/BX+1.0)**2)
      SX = SQRT(VX)
      EY = AY**(1.0/BY) * GAMMA(1.0/BY+1.0) + CY
      VY = AY**(2.0/BY) * (GAMMA(2.0/BY+1.0) - GAMMA(1.0/BY+1.0)**2)
      SY = SQRT(VY)
      EW = EX - EY
      VW = VX + VY
      SW = SQRT(VW)
C
C   P( X .GE. Y )...
C
      PR = XINT( CY, RI(TOL, AY,BY,CY ), P, TOL )
      WRITE(OUTPUT,30) AX,BX,CX, AY,BY,CY, EX,VX,SX, EY,VY,SY, EW,VW,SW
   30 FORMAT( 14H1PARAMETERS...  //,
     1         4X,2HAX,8X,2HBX,8X,2HCX,8X,2HAY,8X,2HBY,8X,2HCY,/,
     2          6F10.5, //,
     3         18X, 4HE(.), 16X, 4HV(.), 16X, 4HS(.), /,
     4         6H X....., 3F20.7, /,
     5         6H Y....., 3F20.7, /,
     6         6H X-Y..., 3F20.7  )
      WRITE(OUTPUT,40) PR
   40 FORMAT( 16H0P( X .GE. Y ) = F10.7 )
      IF( NW .LE. 0 ) GO TO 10
C
C   FIND /NW/ ORDINATES OF DENSITY FROM /-Z/ TO /+Z/ STANDARD DEVIATIONS
C
      UB = RI(TOL, AX,BX,CX)
      UUB = RI(TOL, AY,BY,CY)
      IF( UB .LT. UUB ) UB = UUB
      IF( Z .LE. 0.0 ) Z = 3.0
      WMIN = EW - Z * SW
      WSTEP = (2 * Z * SW ) / ( NW - 1 )
      BL = CX
      WRITE(OUTPUT,50)
   50 FORMAT( 36H1ORDINATES FOR DENSITY OF DIFFERENCE  /,
     1         1H0, 15X, 5HW=X-Y, 17X, 8HORDINATE )
C
C   PRINT ANALYTICAL ORDINATE WHEN X AND Y ARE EXPONENTIAL
C
      LOGIC = 1
      IF( BX .EQ. 1.0 .AND. BY .EQ. 1.0 ) LOGIC = 0
      IF( LOGIC .EQ. 0 ) WRITE(OUTPUT,60)
   60 FORMAT( 1H+, 54X, 11HEXPONENTIAL )
      DO 110 I=1,NW
      W = WMIN + (I-1) * WSTEP
      IF(W .GT. CX - CY) BL = W + CY
      ORD = XINT( BL,UB, H, TOL)
      IF( LOGIC .NE. 0 ) GO TO 100
      IF( W .GT. CX - CY ) GO TO 70
      EXPO = (1.0/(AX+AY))*EXP(( W-CX+CY)/AY)
      GO TO 80
   70 EXPO = (1.0/(AX+AY))*EXP((-W+CX-CY)/AX)
   80 WRITE(OUTPUT,90) W, ORD, EXPO
   90 FORMAT ( 5X, 3F20.6 )
      GO TO 110
  100 WRITE(OUTPUT,90) W, ORD
  110 CONTINUE
      GO TO 10
      END
      FUNCTION FN(X, A, B, C )
C
C   ORDINATE FOR THREE PARAMETER WEIBULL DENSITY FUNCTION
C
      FN = 0.0
      IF( X .LE. C ) GO TO 10
      FN=((B*(X-C)**(B-1.))/A)*EXPF(-(X-C)**B/A)
   10 RETURN
      END
      FUNCTION R(X, A, B, C )
C
C   RELIABILITY FUNCTION FOR THREE PARAMETER WEIBULL DENSITY FUNCTION
C
      R = 1.0
      IF( X .LE. C ) GO TO 10
      R = EXPF(-(X-C)**B/A )
   10 RETURN
      END
      FUNCTION P(X)
C
C   INTEGRATION ARGUMENT FOR DETERMINATION OF P( X .GE. Y )
C
      COMMON AX,BX,CX, AY,BY,CY
      P = FN(X, AY,BY,CY) * R(X, AX,BX,CX)
      RETURN
      END
      FUNCTION H(X)
C
C   INTEGRATION ARGUMENT FOR DENSITY OF W = X - Y
C
      COMMON AX,BX,CX, AY,BY,CY, W
      H = FN(X-W, AY,BY,CY) * FN(X, AX,BX,CX)
      RETURN
      END
      FUNCTION RI( TOL, A,B,C)
C
C   INVERSE OF THREE PARAMETER WEIBULL RELIABILITY FUNCTION
C
      COMMON AX,BX,CX, AY,BY,CY
      RI = C
      T = -A*ALOG(TOL)
      IF( T .LE. 0.0 ) RETURN
      RI = C + T ** (1.0/B)
C
C   SET ABSOLUTE UPPER BOUND FOR /EXPF/ ARGUMENTS IN /FN/ AND /R/
C
      UB  = (600.0 * AX) ** (1.0 / BX) + CX
      UUB = (600.0 * AY) ** (1.0 / BY) + CY
      IF( UB .GT. UUB ) UB = UUB
      IF( RI .GT. UB ) RI = UB
      RETURN
      END
```

REFERENCES

[1] M. Abramowitz and I. A. Stegun, Ed., *Handbook of Mathematical Functions* (NBS Appl. Math Ser. 55). Washington, D.C.: U.S. Government Printing Office, 1968.

[2] J. Berrettoni, "Practical applications of the Weibull distribution," *Industr. Qual. Contr.*, pp. 71-79, Aug. 1964.
[3] B. Carnahan, H. Luther, and J. Wilkes, *Applied Numerical Methods.* New York: Wiley, 1969.
[4] D. Lloyd and M. Lipow, *Reliability: Management, Methods, and Mathematics.* Englewood Cliffs, N.J.: Prentice-Hall, 1962, p. 138.
[5] N. Mann, "Point and interval estimation procedures for two-parameter Weibull and extreme value distributions," *Technometrics*, vol. 10, pp. 231-256, May 1968.
[6] W. Nelson and V. Thompson, "Weibull probability papers," *J. Qual. Technol.*, vol. 3, pp. 45-50, Apr. 1971.
[7] J. Ravenis, "Estimating Weibull distribution parameters," *Electro-Technol.*, pp. 46-54, Mar. 1964.
[8] W. Weibull, "A statistical distribution of wide applicability," *J. Appl. Mech.*, vol. 18, pp. 293-297, Sept. 1951.

Gerald G. Brown was born in Hollywood, Calif., on January 21, 1946.

Mr. Brown has a B.A. in Business and an M.B.A. in Quantitative Methods from California State College, Fullerton. He is now a doctoral candidate in Operations Research at the University of California, Los Angeles. He has been an Assistant Professor of Quantitative Methods at California State College, Fullerton, for three years. His fields of interest include Stochastic Processes and Computer Science, and he has an article in *Operations Research.*

Herbert C. Rutemiller was born in Erie, Pa., on September 5, 1925.

Dr. Rutemiller has a B.S. in Physics and a Ph.D. in Statistics from Case Western Reserve University. He spent 12 years as a Metallurgical Engineer. Since 1962, he has been teaching full time, and he joined California State College, Fullerton, in 1966, where he is currently Professor of Quantitative Methods. His fields of interest include reliability, experimental design, and estimation theory. He has published numerous articles on estimation theory in the *Journal of the American Statistical Association*, *Journal of Business*, and *Operations Research.*

Reprinted by permission from IEEE TRANSACTIONS ON RELIABILITY
Vol. R-22, No. 2, June 1973, pp. 78-82
Copyright 1973, by the Institute of Electrical and Electronics Engineers, Inc.
PRINTED IN THE U.S.A.